Construire un réseau de gaines de climatisation


par Marcel Marlet


 





Ce programme permet de tracer tout un réseau de gaines de traitement d'air, depuis une section principale jusqu'aux diverses ramifications terminant sur des bouches circulaires.





Le programme demande tout d'abord la largeur de la gaine, puis le point de départ.


Ensuite une série d'options permet de développer le réseau avec diverses possibilités:


modifier la largeur de gaine


choisir un décalage (gauche – droit – axial)


clôturer le réseau





La clôture peut se faire avec option:


tronquée: la gaine est arrêtée simplement


bouche : la gaine finit sur une bouche circulaire


dont le diamètre est proportionnel à la largeur de gaine.





Les changements de direction se font avec arrondis.


Un contrôle de la longueur du tronçon en fonction de la largeur de gaine évite des problèmes de raccordement.





Cette routine permet de tracer rapidement une installation et peut servir à des bureaux d'études spécialisés en climatisation.








Le programme se nomme "RESGAINE.LSP". Il est téléchargeable à partir du site 


http://dominique.vaquand.free.fr





Bonne utilisation





Dominique VAQUAND

Liste du fichier RESGAINE.LSP





;============================================================


; RESGAINE.LSP -- Construction d'un réseau de gaines de traitement d'air en 2D


;


;MARLET Marcel - AP-HM


;13005 MARSEILLE


;


;Tel 04 91 38 18 30


;email: MMARLET@mail.ap-hm.fr


;============================================================








;conversion angle degré -> radian


(defun der(a)(* pi (/ a 180.0)))








(defun C:RESGAINE (/ ancmrq ancech pf2 pd pf pr ptint ancpt1 ancpt2   


                     pr1 pr2 r1 r2 r3 r4 dpf4 d13 np nvlep dec


                     dpd1 dpf1 dpd2 dpf2 dpd3 dpf3 dpd4 ancpbox


                     pr12 pr22 d24 rr1 rr2 rr3 rr4 ancaccro)


   (setq ancmrq   (getvar "BLIPMODE")


         ancech   (getvar "CMDECHO")


         ancaccro (getvar "OSMODE")


         ancpbox 3


         olderr *error*


         *error* clerr


   )


   (setvar "CMDECHO" 0)


   (setvar "OSMODE" 0)


   (setvar "PICKBOX" 1)


   (setq pf2 '(0 0))


   (infogn)


   (setvar "BLIPMODE" 0)


   (dessdprt)


   (while (and (not (equal pf2 nil))


               (not (equal pf2 "c"))


               (not (equal pf2 "C"))


          )


          (gnesuite)


   )


   (setvar "BLIPMODE" ancmrq)


   (setvar "CMDECHO" ancech)


   (setvar "OSMODE" ancaccro)


   (setvar "PICKBOX" ancpbox)


   (setq *error* olderr)


   (princ)


)





(defun infogn()


   (if (= init nil)


       (while (= init nil)


              (setq init (getdist "\nLargeur de la gaine : ")


              )


       )


       (setq init (getdist (strcat "\nLargeur de la gaine < " (rtos dist)


                                   " > : "


                           )


                  )


       )


   )


   (if (= init nil)


       (setq init dist)


       (setq dist init)


   )


)





;================================


;                       Démarrage du réseau


;================================





(defun dessdprt()


   (setq pd (getpoint "\nPoint de départ : ")


         pf (getpoint pd "\nExtrémité : ")


   )


   (command "LIGNE" (setq dpd1


                          (polar pd (+ (angle pd pf)(der 90)) (/ dist 2))


                    )


                    (setq dpf1


                          (polar pf (+ (angle pd pf)(der 90)) (/ dist 2))


                    )


                    ""


            "LIGNE" (setq dpd2


                          (polar pd (- (angle pd pf)(der 90)) (/ dist 2))


                    )


                    (setq dpf2


                          (polar pf (- (angle pd pf)(der 90)) (/ dist 2))


                    )


                    ""


   )


   (setq ancpt1 dpd1 ancpt2 dpd2)


)





;=================================


;                          Diverses options


;=================================





(defun gnesuite()


   (initget "C L")


   (setq pf2 (getpoint pf "\nLargeur/Clore/<Point suivant> : "))


   (if (and (not (equal pf2 nil))


            (not (equal pf2 "c"))


            (not (equal pf2 "C"))


            (not (equal pf2 "l"))


            (not (equal pf2 "L"))


       )


       (if (<= (distance pf pf2) (* dist 2))


           (list (while (<= (distance pf pf2) (* dist 2))


                     (setq pf2 


                       (getpoint pf "\nDistance insuffisante ! Autre Point:")


                     )


                 )


                (dessgne)


           )


           (dessgne)


       )


       (if (or (equal pf2 nil)


               (equal pf2 "c")


               (equal pf2 "C")


           )


           (gneclore)


           (gainrac)


       )


   )


)





;=================================


;                         Dessin du tronçon


;=================================





(defun dessgne()


   (setq dpd3   (polar pf (+ (angle pf pf2)(der 90)) (/ dist 2))


         dpd4   (polar pf (- (angle pf pf2)(der 90)) (/ dist 2))


         dpf3   (polar pf2 (+ (angle pf pf2)(der 90)) (/ dist 2))


         dpf4   (polar pf2 (- (angle pf pf2)(der 90)) (/ dist 2))


   )      


   (if (equal dpd3 dpf1)  ;  pas de changement de direction





       (list (command "LIGNE" dpf1 dpf3 ""


                      "LIGNE" dpf2 dpf4 ""


             )


             (setq ancpt1 dpd3 ancpt2 dpd4)


       )


       (list (setq r1 (polar dpf1 (+ (angle pd pf)(der 180)) dist)


                   r2 (polar dpf2 (+ (angle pd pf)(der 180)) dist)


                   r3 (polar dpd3 (angle pf pf2) dist)


                   r4 (polar dpd4 (angle pf pf2) dist)


                   d13 (distance r1 r3)


                   d24 (distance r2 r4)


             )


                   ;  construction du raccord





             (command "EFFACER" (entlast) ""


                      "EFFACER" (entlast) ""


                      "LIGNE" ancpt1 r1 ""


                      "LIGNE" ancpt2 r2 ""


             )


             (setq ptint (inters r1 r2 r3 r4 nil))


             (if (< d24 d13)


                 (command "ARC" r4 "c" ptint r2 


                          "ARC" r3 "c" ptint r1


                 )


                 (command "ARC" r2 "c" ptint r4 


                          "ARC" r1 "c" ptint r3


                 )


             )


             (if (> dist 0.20)   ;  traçage des liaisons tronçon-raccord


                 (list (setq rr1 (polar r1 (+ (angle pd pf)(der 90)) 


                                        (* 0.06 dist))


                             rr2 (polar r2 (- (angle pd pf)(der 90)) 


                                        (* 0.06 dist))


                             rr3 (polar r3 (+ (angle pf pf2)(der 90)) 


                                        (* 0.06 dist))


                             rr4 (polar r4 (- (angle pf pf2)(der 90)) 


                                        (* 0.06 dist))


                       )


                       (command "LIGNE" rr1 rr2 ""


                                "LIGNE" rr3 rr4 ""


                       )


                 )


             )


             (command "LIGNE" r3 dpf3 ""


                      "LIGNE" r4 dpf4 ""


             )


             (setq ancpt1 r3 ancpt2 r4)


       )


   )


   (setq pd pf pf pf2 dpf1 dpf3 dpf2 dpf4)     ;   transfert de variables


)





;===========================================


;                             Changement   de   largeur


;===========================================





(defun gainrac()


   (setq dec 0


         al  dist


         init (getdist (strcat "\nAncienne largeur < " (rtos dist)


                               " > Entrer nouvelle : "


                       )


              )


   )


   (if (or (= init nil)


           (= init dist)


       )


       (setq init dist


             nvlep dist


       )


       (setq dist init)


   )


   (if (= nvlep dist)


       (gnesuite)


       (decchoix)


   )


)





;===================================================


;                           Choix   du   décalage ( réduction )


;===================================================





(defun decchoix()


   (initget "G D S nil")


   (setq dec (getkword "\nDécalage/départ - Gauche/Droite/<Symétrique> : "))


   (if (or (equal dec "g")


           (equal dec "G")


       )


       (gchreduc)


   )


   (if (or (equal dec "d")


           (equal dec "D")


       )


       (drtreduc)


   )


   (if (or (equal dec "s")


           (equal dec "S")


           (equal dec nil)


       )


       (desreduc)


   )


)





;========================================


;                                Réduction   gauche


;========================================





(defun gchreduc()


   (setq lt (getdist pf "\nLongueur du Tronçon : "))


   (if (< lt (* 2.25 dist))


       (while (< lt (* 2.25 dist))


              (setq lt 


                    (getdist pf "\nDistance insuffisante ! Autre Point:")


              )


       )


   )


   (setq pf2 (polar pf (angle pd pf) lt)


         pr  (polar pf (angle pd pf) dist)


   )      


   (if (> al dist)


       (setq np (/ (- al dist) 2))


       (setq np (/ (- dist al) 2))


   )


   (setq pf2 (polar pf2 (- (angle pd pf)(der 90)) np)      


         pr  (polar pr (- (angle pd pf)(der 90)) np)


         pr1 (polar pr (+ (angle pd pf)(der 90)) (/ dist 2))


         pr2 (polar pr (- (angle pd pf)(der 90)) (/ dist 2))


         pr12 (polar pf2 (+ (angle pd pf)(der 90)) (/ dist 2))


         pr22 (polar pf2 (- (angle pd pf)(der 90)) (/ dist 2))


   )


   (command "LIGNE" dpf1 


                    pr1


                    pr2


                    dpf2


                    dpf1


                    ""


            "LIGNE" pr1


                    pr12


                    ""


            "LIGNE" pr2


                    pr22


                    ""


   )


   (setq pd pr pf pf2 dpf1 pr12 dpf2 pr22 ancpt1 pr1 ancpt2 pr2)


)





;======================================


;                                Réduction   droite


;======================================





(defun drtreduc()


   (setq lt (getdist pf "\nLongueur du Tronçon : "))


   (if (< lt (* 2.25 dist))


       (while (< lt (* 2.25 dist))


              (setq lt 


                    (getdist pf "\nDistance insuffisante ! Autre Point:")


              )


       )


   )


   (setq pf2 (polar pf (angle pd pf) lt)


         pr  (polar pf (angle pd pf) dist)


   )      


   (if (> al dist)


       (setq np (/ (- al dist) 2))


       (setq np (/ (- dist al) 2))


   )


   (setq pf2 (polar pf2 (+ (angle pd pf)(der 90)) np)      


         pr  (polar pr (+ (angle pd pf)(der 90)) np)


         pr1 (polar pr (+ (angle pd pf)(der 90)) (/ dist 2))


         pr2 (polar pr (- (angle pd pf)(der 90)) (/ dist 2))


         pr12 (polar pf2 (+ (angle pd pf)(der 90)) (/ dist 2))


         pr22 (polar pf2 (- (angle pd pf)(der 90)) (/ dist 2))


   )


   (command "LIGNE" dpf1 


                    pr1


                    pr2


                    dpf2


                    dpf1


                    ""


            "LIGNE" pr1


                    pr12


                    ""


            "LIGNE" pr2


                    pr22


                    ""


   )


   (setq pd pr pf pf2 dpf1 pr12 dpf2 pr22 ancpt1 pr1 ancpt2 pr2)


)





;=============================================


;                       Réduction   axiale  ( symétrique )


;=============================================





(defun desreduc()


   (setq lt (getdist pf "\nLongueur du Tronçon : "))


   (if (< lt (* 2.25 dist))


       (while (< lt (* 2.25 dist))


              (setq lt 


                    (getdist pf "\nDistance insuffisante ! Autre Point:")


              )


       )


   )


   (setq pf2 (polar pf (angle pd pf) lt)


         pr  (polar pf (angle pd pf) dist)


         pr1 (polar pr (+ (angle pd pf)(der 90)) (/ dist 2))


         pr2 (polar pr (- (angle pd pf)(der 90)) (/ dist 2))


         pr12 (polar pf2 (+ (angle pd pf)(der 90)) (/ dist 2))


         pr22 (polar pf2 (- (angle pd pf)(der 90)) (/ dist 2))


   )


   (command "LIGNE" dpf1 


                    pr1


                    pr2


                    dpf2


                    dpf1


                    ""


            "LIGNE" pr1


                    pr12


                    ""


            "LIGNE" pr2


                    pr22


                    ""


   )


   (setq pd pr pf pf2 dpf1 pr12 dpf2 pr22 ancpt1 pr1 ancpt2 pr2)


)





;=======================================


;                          Clôture   du   réseau


;=======================================





(defun gneclore()


   (initget "T B nil")


   (setq term (getkword "\nBouche/<Tronqué> : "))


   (if (or (equal term "t")


           (equal term "T")


           (equal term nil)


       )


       (command "LIGNE" dpf1


                        dpf2


                        ""


       )


       (command "CERCLE" pf (* dist 1.5)


                "AJUSTER" dpf2 dpf1 ""


                          (polar pf (angle pf pd) (* dist 1.5))


                          ""


                "ARC" "c"


                      pf


                      dpf2


                      dpf1


       )





   )    


);fin de defun gneclore()





(princ "\nRESGAINE .............commande chargée")


(princ)





;........................ Fin de RESGAINE.LSP











